
Abstract
A recent paper by Graham Hutton and Diana Fulger (‘Reasoning
about Effects: Seeing the Wood through the Trees’, in the prepro-
ceedings of Trends in Functional Programming, 2008) addresses
the problem of reasoning about effectful functional programs, in-
troducing a relabelling function on binary trees as a representative
illustration. The example is a very fruitful one, but we argue that
their approach is less effective than it might be, because they miss
two opportunities for higher-level reasoning: abstraction from the
particular kinds of effect (the choice of monad) and from the pattern
of recursion (the flow of computation). We present an alternative
approach using properties of idiomatic traversals, which cleanly
separate the twin concerns of the kinds of effect and the pattern of
recursion. In particular, we approach the problem by considering
its inverse; and we argue that this is an important approach which
has so far been missing from discussions of idiomatic traversal.

Life can only be understood backwards;
but it must be lived forwards.

— Søren Kierkegaard

1. Introduction
How does the presence of effects change our ability to reason about
functional programs? The question has been around for years, but
if you add ‘equationally’ after ‘reasoning’ and ‘lazy’ before ‘func-
tional’, there doesn’t seem to have been much written that will help
Jane Programmer in her daily task of constructing Haskell code.
One answer to the question is to say that it shouldn’t. Instead the
advice is to reason about pure functional programs, and then me-
chanically translate the results into impure ones, presumably for
efficiency reasons. But that only works for programs that have pure
counterparts. Another answer is given in a recent paper by Hut-
ton and Fulger [5]. They study a deliberately simple example, that
of labelling binary trees. Their objective was to find some way of
reasoning that the labelling, expressed using the state monad, gen-
erates distinct labels.

[copyright notice will appear here]

While we appreciate that one good example is worth 1000 the-
ories, we believe that Hutton and Fulger’s development is some-
what unsatisfactory, and for two separate reasons. Firstly, despite
noting that ‘unfortunately, it still remains standard practice to just
expand out the basic effectful operations when trying to reason’,
their proof does exactly that—they reduce stateful computations to
pure functions that accept and return a state. (This expansion is
encapsulated in separate lemmas, but that is a surface textual ab-
straction rather than a deeper mathematical one.) This means that
the effort invested in reasoning about a stateful program cannot be
reused if the program is modified to exploit other kinds of effects.
Moreover, it is unclear whether the approach can be used at all for
effects such as input/output, that don’t have pure equivalents. It’s
the same criticism that we gave for the first answer.

Secondly, they resort to explicit recursion in implementing the
relabelling function, which consequently forces them to use explicit
induction in reasoning about it. They conclude that ‘it would be
interesting to see if there is any benefit to be gained from using a
more structured approach to recursion’.

We agree. Our intention is to show that there is a great benefit
to be gained from using structured recursion; indeed, the essence
of this particular problem is precisely the structure of the recursion.
In particular, the tree labelling problem is an elegant example of
an idiomatic traversal. Moreover, stating the problem in terms of a
structured recursion operation such as idiomatic traversal forms a
clear separation of concerns between the ‘plumbing’ aspects of the
problem (which only concern the traversal) and the ‘computational’
aspects (which concern the particular effects). The major part of the
reasoning can be performed independently of the effects, and so is
applicable also to other kinds of effect.

2. Tree labelling
Here is the tree datatype in question:

data Tree α = Tip α | Bin (Tree α) (Tree α)

Hutton and Fulger offer several related labelling functions, but the
one we will adopt in the first instance is subtly different, and we will
come back to why later on. In our version, tree labelling annotates
a tree with additional elements drawn from an infinite stream, the
stream being threaded through the computation via the state monad:

label :: Tree α → State [β ] (Tree (α,β ))
label (Tip a) = do {(b : y)← get;put y;return (Tip (a,b))}
label (Bin u v) = do {u′← label u;v′← label v;return (Bin u′ v′)}

The essential property that Hutton and Fulger wished to prove is
that tree elements are annotated with distinct labels. Because our

version is polymorphic in the label type, we cannot talk about ‘dis-
tinctness’; instead, we require that the labels used are drawn with-
out repetition from the given stream—consequently, if the stream
has no duplicates, the labels will be distinct. In turn, this is a corol-
lary of the following property: the sequence of labels used to label
the tree, when prepended back on to the stream of unused labels,
forms the original input stream of labels. The function labels ex-
tracts the annotations:

labels :: Tree (α,β )→ [β ]
labels (Tip (a,b)) = [b]
labels (Bin u v) = labels u++ labels v

The crucial property to be proved is that

runState (label t) xs = (u,ys) ⇒ labels u++ ys = xs

3. Idiomatic traversals
We argue that the labelling problem is inherently one about effectful
iterations over data structures; moreover, as far as possible, any rea-
soning should be abstracted both from the specific computational
effect and from the shape of the data structure being iterated over.
We encapsulate the effects as idioms [7], rather than the more famil-
iar monads; as we shall see, they have better compositional proper-
ties, which will prove useful in what follows. And we encapsulate
the shape of the data structure in terms of idiomatic traversals, also
introduced by McBride and Paterson, and studied in more depth in
[3]. We briefly introduce those two notions here, and show their
immediate connection to the tree labelling problem, before moving
on to a review of the laws of idiomatic traversals.

First of all, idioms are functors with some additional structure:

class Functor m⇒ Idiom m where
pure :: α → m α

(~) :: m (α → β )→ m α → m β

satisfying four laws:

pure id~ma = ma
pure (◦)~mf ~mg~ma = mf ~ (mg~ma)
pure f ~pure a = pure (f a)
mf ~pure a = pure (λ f → f a)~mf

Every monad is an idiom. For example, the state monad is an idiom,
with operations defined as follows:

instance Idiom (State s) where
pure a = return a
mf ~ma = do {f ← mf ;a← ma;return (f a)}

That is, pure computations coincide with the ‘return’ of the monad,
and idiomatic application yields the effects of evaluating the func-
tion before the effects of evaluating the argument (as captured by
the function ap in the Haskell libraries). But we could also have
defined

mf ~ma = do {a← ma; f ← mf ;return (f a)}
evaluating arguments before functions. So, monads translate into
idioms in two reasonable and dual ways. On the other hand there
are idioms that do not arise from monads. Any constant functor
returning a monoidal type yields a ‘phantom’ idiom, in which the
type parameter is a phantom type; pure computations yield the
neutral element of the monoid, and idiomatic application reduces
to the binary operator:

newtype K α β = K {unK :: α }
instance Monoid α ⇒ Idiom (K α) where

pure a = K mempty
K a~K b = K (mappend a b)

Next, a traversable datatype according to [7] is one that supports
two inter-definable functions traverse and dist. The former applies
an effectful body to every element of a data structure, collecting all
the effects in order. In the case of a monadic idiom and when the
datatype is lists, this reduces to the monadic map mapM. The func-
tion dist takes a data structure of computations to a computation
yielding a data structure, chaining together all the effects to dis-
tribute the data structure over the idiom. For monadic idioms and
lists, this is the function sequence.

class Functor t⇒ Traversable t where
traverse :: Idiom m⇒ (α → m β )→ t α → m (t β )
traverse f = dist ◦ fmap f
dist :: Idiom m⇒ t (m α)→ m (t α)
dist = traverse id

For example, trees form a traversable datatype; traversal of a tip
involves visiting its label, and one possible traversal of a binary
node involves traversing the left subtree before the right:

instance Traversable Tree where
traverse f (Tip a) = pure Tip~ f a
traverse f (Bin u v) = pure Bin~ traverse f u~ traverse f v

Another choice is to traverse the right subtree before the left one.
And these are not the only two possible definitions of traverse. We
will come back to this important point later on. Tree labelling can
be formulated as a tree traversal, using the monadic idiom arising
from the state monad:

label :: Tree α → State [β ] (Tree (α,β ))
label = traverse adorn

The body of the traversal consumes a single label from the stream,
using it to adorn a single tree element:

adorn :: α → State [β ] (α,β )
adorn a = do {(b : y)← get;put y;return (a,b)}

The crunch question is: Can we reason about labelling simply
by reasoning about the properties of traverse? The answer is a
qualified ‘yes’, but we won’t mention the qualification just yet.

4. Properties of idioms and traversal
We briefly remind ourselves of some properties of idioms and of
idiomatic traversal, following [7] and [3]. First off, idioms compose
nicely, in the exactly the way monads don’t:

data C m n α = C {unC :: m (n α)}
instance (Idiom m, Idiom n)⇒ Idiom (C m n) where

pure a = C (pure (pure a))
C mnf ~C mna = C (pure (~)~mnf ~mna)

We can introduce a composition operator for idiomatic computa-
tions:

(�) :: (Idiom m, Idiom n)⇒
(β → n γ)→ (α → m β )→ α → C m n γ

g� f = C ◦ fmap g◦ f

Of course, there’s an identity idiom too:

newtype I α = I {unI :: α }
instance Idiom I where

pure a = I a
I f ~ I a = I (f a)

We require traversal to respect the compositional structure of id-
ioms. That is to say, for a pure function f :: α → β we require that

traverse (I ◦ f ) = I ◦ fmap f

and, for effectful bodies f :: α → m β and g :: β → n γ , that

traverse (C ◦ fmap g◦ f ) = C ◦ fmap (traverse g)◦ traverse f

Using idiom composition, the latter equation is equivalent to

traverse (g� f ) = traverse g� traverse f

which can be seen as a fusion law for idiomatic traversals.
The next property is that traversal should be natural in the idiom.

By definition, an idiom morphism φ :m .→ n from idiom m to idiom
n is a polymorphic function φ : m α → n α such that

φ (pure a) = pure a
φ (mf ~ma) = (φ mf )~ (φ ma)

Given an idiom morphism φ we expect the two laws

φ ◦ traverse f = traverse (φ ◦ f )
φ ◦dist = dist ◦ fmap φ

to hold (each of which implies the other). In particular, pure◦unI :
I .→ m is an idiom morphism; as a consequence, we get the purity
law that traversal with pure is itself just pure:

traverse pure = pure

Also, when m is the idiom arising from a commutative monad
and join :: m (m α) → α is monad multiplication, we have that
join ◦ unC : C m m .→ m is an idiom morphism; as a consequence,
we get a special fusion law for monadic traversals in a common
commutative monad:

traverse g• traverse f = traverse (g• f )

where • is Kleisli composition:

(g• f ) a = do {b← f a;g b}

5. Composing idiomatic traversals
How might we set about proving the property at the end of Section 2
stating the correctness of tree labelling? The first observation is
that the two functions label and labels are written in quite different
styles—the first as an effectful traversal, and the second as a pure
function—and so their combination requires flattening the ‘state’
abstraction (via the runState function). Unifying the two styles
entails either writing label in a pure style (which is possible, but
which amounts to falling back to first principles), or writing labels
in an effectful style. Hutton and Fulger took the former approach;
we take the latter.

We can extract the labels from an annotated tree as another
stateful traversal, operating on the same state type of streams of
labels. The body of the traversal strips the label from a single
labelled element, and returns it to the stream of unused labels:

strip :: (α,β )→ State [β ] α

strip (a,b) = do {y← get;put (b : y);return a}
We can then traverse with this body, stripping all the labels off a
labelled tree.

unlabel :: Tree (α,β )→ State [β ] (Tree α)
unlabel = traverse strip

One might expect that label, which adorns a tree with labels from
a stream, and unlabel, which strips those labels off again, and
returns them to the stream, would be in some sense each other’s
inverses. More precisely, suppose that m and n are idioms, and that
f :: α → m β and g :: β → n α . We can reason

traverse g� traverse f
= [[ traversal respects composition ]]

traverse (g� f )

= [[ suppose g� f = pure ]]
traverse pure

= [[ purity law ]]
pure

This reasoning would be applicable in our situation, if we could
show that

strip�adorn = pure

But this property doesn’t hold; the two sides are observably differ-
ent. It turns out that � is a rather odd kind of ‘translucent compo-
sition’: in the composition g� f , the intermediate value returned
by f is hidden, but the effects of f are not. (Concretely, defining
observation function h by

h mna = pure (const ())~unC mna

the reader may confirm that

h ((strip�adorn) a) = do {(b : y)← get;put y;return ()}

whereas details are
in Ap-
pendix A.1h (pure a) = return ()

The two right-hand sides are clearly different, and so strip�adorn
and pure must differ too.)

Let’s try again, this time with Kleisli composition and monads.
Both strip and adorn use the same monadic idiom and this time we
do have

strip•adorn = return

Now we can reason (but spot the bug!)

traverse g• traverse f
= [[ traversal respects Kleisli composition ]]

traverse (g• f )
= [[ assuming g• f = return ]]

traverse return
= [[ purity law ]]

return

Did you spot the bug? Traversal respects Kleisli composition only
for commutative monads and the state monad is the epitome of non-
commutativity. (Incidentally, Gibbons and Oliveira [3] conjectured
that traversal respects Kleisli composition see Ap-

pendix A.2
traverse g• traverse f = traverse (g• f )

even for a non-commutative monad, provided that f and g them-
selves commute, but that conjecture is false. It is not too difficult to
construct a counterexample with the state monad.)

6. Knowing it forwards and backwards
Perhaps the reader has been ahead of us, and seen already that the
two attempts above cannot possibly work. Both label and unlabel
traverse the tree in the same direction, and so the supposedly in-
verse effects of unlabel take place in the wrong order, and the two
functions can’t possibly be inverse. After all, if you put your socks
and then your shoes in in the morning, then in the evening you take
off your shoes and then your socks.

Another property enjoyed by idioms and not in general by mon-
ads is that, for each idiom m, there is a corresponding ‘backwards’
idiom B m with the effects sequenced in the opposite order.

newtype B m α = B {unB :: m α }
instance Idiom m⇒ Idiom (B m) where

pure a = B (pure a)
B mf ~B ma = B (pure (flip ($))~ma~mf )

Reversing the order of effects allows us to define backwards traver-
sal (one meaning of ‘recurse’ is ‘go backwards’, from the Latin
‘recurrere’):

recurse :: (Idiom m,Traversable t)⇒ (α → m β )→ t α → m (t β )
recurse f = unB◦ traverse (B◦ f )

In other words, recurse is an instance traverse, visiting every ele-
ment of a data structure, but collecting the effects in the opposite
order; for example, with the earlier definition of traverse on binary
trees, we have:proof is in

Figure A.1
recurse f (Bin u v) = pure (flip Bin)~ recurse f v~ recurse f u

In particular, we can define:

unlabel = recurse strip

Of course, by definition B : m .→ B m is a natural transformation
between two idioms. But it is not an idiom morphism; there is in
general no relationship between B mf ~ B ma and B (mf ~ ma).
However, the composition B ◦ B : m .→ m is an idiom morphism,
expressing the observation that reversal of the order of effects is an
involution. As a consequence, we can derive a dual characterizationproof is in

Figure A.2 of traverse in terms of reverse:

traverse f
= [[ B is an isomorphism ]]

unB◦unB◦B◦B◦ traverse f
= [[ B◦B is an idiom morphism ]]

unB◦unB◦ traverse (B◦B◦ f )
= [[ definition of recurse ]]

unB◦ recurse (B◦ f )

Now, we can say something about Kleisli compositions of
traverse and recurse. We might hope for a general composition
law of the form

recurse f • traverse g = traverse (f •g) -- invalid!

but such a hope is still forlorn—the effects of f happen in opposite
orders on the two sides of the equation. Besides, this law is sus-
picious on grounds of symmetry alone: why traverse on the right-
hand side, rather than recurse? However, we can impose the law
in the rather special case that the traversal bodies are inverse, that
is, f • g = return. Then the asymmetry disappears, because of the
purity law:

traverse return = return = recurse return

That is, we expect recurse f and traverse g to be inverses whenever
f and g are:

f •g = return ⇒ recurse f • traverse g = return

We impose this inverse traversal law as an additional axiom; then
we have

unlabel• label
= [[ definitions ]]

recurse strip• traverse adorn
= [[ given that strip•adorn = return ]]

return

Finally our proof obligation is complete, and everything in the
garden is rosy.

Let’s now return to a point made in Section 2 about tree la-
belling, and confess that our version of the problem is not the one
Hutton and Fulger actually chose. The calculation above, brief and
attractive as it is, depends on being able to invert the original func-
tion. What if the function is not invertible? For instance, Hutton and
Fulger originally presented the labelling problem non-injectively—

in terms of stateful computations working on an integer state, and
replacing rather than annotating the existing tree labels:

labeln :: Tree α → State Int (Tree Int)
labeln = traverse fresh
fresh :: α → State Int Int
fresh a = do {n← get;put (n+1);return n}

Presented in this way, the problem becomes one of showing that

runState (labeln t) m = (u,n) ⇒ nodups (contents u)

where the predicate nodups tests whether a list has no duplicates,
and contents is a traversal in the monoidal idiom arising from the
list monad:

contents :: Traversable t⇒ t α → [α ]
contents = unK ◦ traverse (λa→ K [a])

But the traversal labeln is not invertible, so at first blush reasoning
expressed in terms of backwards traversal does not seem to be very
helpful. All is not lost, however, because we can redefine labeln so
that it becomes invertible. Define new versions of adorn and strip
by

adorn :: α → State ([α ], [β ]) β

adorn a = do {(x,b : y)← get;put (a : x,y);return b}
strip :: β → State ([α ], [β ]) α

strip b = do {(a : x,y)← get;put (x,b : y);return a}
The function adorn replaces the original labels with new ones but
remembers the old ones, while strip does the reverse. We have
strip•adorn = return, as before. Secondly, define new versions of
label and unlabel by

label = traverse adorn
unlabel = recurse strip

Then, as before, unlabel• label = return. Moreover,

runState (labeln t) m = runState (label t) ([ ], [m . .])

and our reasoning remains intact.

7. Discussion
We have shown a proof of correctness of a simple effectful pro-
gram operating on a recursive data structure, using techniques that
are modular in both the nature of the effects and the shape of the re-
cursive datatype. We were somewhat surprised that the correctness
arguments did not simply fall out straight away; evidently they de-
pend on properties of traversal that were not immediately obvious
to us (indeed, we overlooked them in earlier related work [3]).

The core of our proof is the relationship between forwards and
backwards traversal. The two functions traverse and recurse form
a very natural pair. Moreover, their interaction does not seem to us
at present to be something that arises from properties of traverse
alone; it has to be stated as a separate axiom. We therefore pro-
pose that recurse should be elevated to first-class status as a third
method in the Traversable class—following the principle that the
most appropriate place in which to state an axiom constraining
implementations of an abstract operation is on the type class of
which the operation is a member. The recurse operation can be
given a default definition in terms of traverse, so this is no addi-
tional burden on the programmer; but if an independent definition
is given, it is the programmer’s responsibility to ensure that the law
B◦ recurse f = traverse (B◦ f ) is satisfied.

Of course, not all effectful programs are instances of traverse,
so the techniques we have discussed in this paper will not always
be applicable. Nevertheless, it seems to us that it is important to
exploit whatever patterns of recursion we find. Similarly, not all

properties of effectful programs can be analyzed independently of
the nature of the effects; sometimes the specific effect is important,
and must be taken into account. But still, we feel that it is important
to distinguish the general from the particular, and worthwhile to
see what can be said about the general case. The curious reader
may wish to read our related paper [4] about what more can be said
about other patterns of control, and specific classes of effects.

There is also an intriguing connection with the factorization of
datatypes into shape and contents, as in Jay’s notion of ‘shapely
over lists’ datatypes [6] and Abbott’s ‘container datatypes’ [1].
We already saw in Section 5 a definition of contents in terms of
traversal in the constantly-lists idiom:

contentst :: Traversable t⇒ t α → [α ]

(explicitly annotating with a superscript to denote the particular
traversable datatype). Defining shape is even easier: it is a traversal
in the identity idiom, that is, a map:

shapet :: Functor t⇒ t α → t ()
shapet = fmap (const ())

Rypáček [8] has observed that traversal should be independent
of the shape of a data structure, in a certain sense: if the natural
transformation ψ : t .→ u preserves contents:

contentsu ◦ψ = contentst

then traversal respects ψ:

fmapm ψ ◦ traverset f = traverseu f ◦ψ

Assuming this additional axiom, one can reason about tree rela-
belling in terms of simpler list relabelling [2]. What is more, we
can specify datatype-generic reversal reverset in terms of shape and
contents—the reverse of a data structure has the same shape and re-
versed contents:

shapet (reverset x) = shapet x
contentst (reverset x) = reverse (contentst x)

Then we might expect that backwards traversal over the reverse
data structure corresponds to forwards traversal over the original:

fmapm reverset (recurse f x) = traverse f (reverset x)

It remains to be seen whether the inverse traversal property of
traverse and recurse is a consequence of some deeper structure that
is yet to be discovered. In particular, symmetry suggests a dual law

f •g = return ⇒ traverse f • recurse g = return

We conjecture that this is equivalent to the inverse traversal axiom
used in Section 6, but we have not yet managed to prove this fact.
Similarly, perhaps the inverse traversal property can be deduced by
looking at datatype-generic reversal.

Appendix
This appendix contains justifications for some of the assertions
made in the body of the paper, as an aid to reviewers. We suggest
that it should not be formally published, even if the paper is ac-
cepted; but perhaps it should form supplementary materials in the
ACM Digital Library. At least then it would provide a venue for
one of us (JG) to confess to earlier unsuccessful conjectures [3].

A.1 Translucent composition
In Section 5, we claimed that the idiomatic composition strip�
adorn was not equal to simply pure in the composite idiom. More
generally, � is a kind of ‘transparent composition’, in the sense
that it doesn’t fully hide the intermediate stage of a composite
computation f � g: it is possible to recover the effects of g alone,
after the fact. Specifically, we have:

pure (const ())~unC ((g� f ) a)
= [[ definition of �; unC ◦C = id ]]

pure (const ())~ fmap g (f a)
= [[ naturality—see below ]]

pure (const ()◦g)~ f a
= [[ const ]]

pure (const ())~ f a

The naturality property cited is that

pure g~ fmap f ma = pure (g◦ f )~ma

which is obtained by specializing f = id, p = q◦ f ′, and h = λg→
(purem g~) in the free theorem

f ◦p = q◦ f ′ ⇒ fmapm f ◦h p = h q◦ fmapm f ′

for a function h of type (α→ β )→ (m α→m β ). This calculation
shows that, from the application (g� f ) a of the idiomatic compo-
sition g� f to a, it is possible to extract the effect of f a alone—
albeit not the result, which is discarded above by the application of
const ()—regardless of what g does. In particular, since adorn has a
non-trivial effect on the state, strip�adorn is distinguishable from
pureC (State [β ]) (State [β ])—even if strip subsequently undoes the ef-
fects, the intermediate state betraying the effects is observable.

A.2 Traversal fusion for non-commutative monads
At the end of Section 5, we mentioned the conjecture by Gibbons
and Oliveira [3, Section 5.4] that

traversem g• traversem f = traversem (g• f )

even for a non-commutative monad m, provided that f and g them-
selves commute. In fact, that conjecture is false, as we show here.

Let f :: α → m β and g :: β → m γ . For a counter-example, we
first need f ,g that commute; for that to make sense, we certainly
need α = β = γ . We will use the state monad m = State s, with
s being streams of labels as throughout; we will also make α =
β = γ = s (so that each data structure element is itself a stream of
labels). Define:

f ,g :: [α ]→ State [α ] [α ]
f x = State (λ (b : y)→ (b : x,y))
g (b : x) = State (λy→ (x,b : y))

Then clearly

f •g = g• f = return

as required for the premise. But with the usual left-to-right traversal
over pairs, we have

traverse f (x,y) = State (λ (a : b : z)→ ((a : x,b : y),z))
traverse g (a : x,b : y) = State (λ z→ ((x,y),b : a : z))

Therefore traverse g• traverse f is not return, since it does not leave
the state unchanged—it swaps the first two elements of the stream
around.

A.3 Linear traversals
We have identified various levels of naturality property for traverse
(and hence for dist and recurse too). Of course, traversal should be
natural in the elements of the data structure being traversed:

fmapm (fmapt h)◦ traversem f = traversem (fmapm h◦ f )

We also required traversal to be natural in the idiom; for idiom
morphism φ : m .→ n,

φ ◦ traversem f = traversen (φ ◦ f )

Finally, we required traversal to respect the compositional structure
of idioms: for a pure function f :: α → β we expect that

traverseI (I ◦ f ) = I ◦ fmapt f

and, for effectful bodies f :: α → m β and g :: β → n γ , that

traverseC m n (g� f ) = traversen g� traversem f

Gibbons and Oliveira [3] showed that these conditions ruled out
various kinds of bogus traversal, including traversals that skip el-
ements or that change the shape of the data structure. However, it
wasn’t clear whether the conditions ruled out traversals that dupli-
cate elements, such as the following bogus traversal of binary trees
that visits every tip twice:

instance Traversable Tree where
traverse f (Tip a) = pure (const Tip)~ f a~ f a
traverse f (Bin u v) = pure Bin~ traverse f u~ traverse f v

It still isn’t clear; but recently, Jaskelioff and Rypacek [9] have of-
fered some evidence—albeit not yet a proof—to support the con-
jecture that all such duplicitous traversals are ruled out by respect
for the compositional structure of idioms. They present an example
of a duplicitous traversal that does not respect composition, using
the list functor three times (traversal of a list, with a body in the
composite idiom arising from the list monad composed with itself).
A perhaps more perspicuous example uses three different functors:
a duplicitous traversal of the identity functor

instance Traversable I where
traverse f (I x) = pure (const I)~ f x~ f x

and traversal bodies in the idioms arising from the exception and
list monads:

f :: Integer→Maybe Integer
f n = if odd n then Just n else Nothing
g :: Integer→ [Integer ]
g n = [n,n+1]

Then we have

traverse (g� f ) (I 0)
= C [Nothing,Nothing,Nothing,Just (I 1)]

(traverse g� traverse f ) (I 0)
= C [Nothing,Just (I 1),Nothing,Just (I 1)]

These plainly differ, so this traversal does not respect idiom com-
position. But it remains an open question whether any such duplic-
itous traversal will be similarly betrayed.

Additional references
[9] Mauro Jaskelioff and Ondřej Rypáček. An investigation of the laws

of traversals. In James Chapman and Paul Blain Levy, editors,
Mathematically Structured Functional Programming, volume 76 of
EPTCS, pages 40–49, 2012.

Here, we show that

recurse f (Bin u v) = pure (flip Bin)~ recurse f v~ recurse f u

given the natural definition of traverse for binary trees, and the characterization of recurse in terms of traverse:

traverse f (Bin u v) = pure Bin~ traverse f u~ traverse f v
recurse f = unB◦ traverse (B◦ f )

To be explicit, we write ‘pureB’ and ‘~B’ for pure and ~ in the backwards idiom.

recurse f (Bin u v)
= [[ definition of recurse in terms of traverse ]]

unB (traverse (B◦ f ) (Bin u v))
= [[ definition of traverse ]]

unB (pureB Bin~B traverse (B◦ f ) u~B traverse (B◦ f ) v)
= [[ pure in a backwards idiom; recurse in terms of traverse ]]

unB (B (pure Bin)~B B (recurse f u)~B B (recurse f v))
= [[ ~ in a backwards idiom ]]

unB ((B (pure (flip ($))~ recurse f u~pure Bin))~B B (recurse f v))
= [[ ~ in a backwards idiom again ]]

unB (B (pure (flip ($))~ recurse f v~ (pure (flip ($))~ recurse f u~pure Bin)))
= [[ unB◦B = id ]]

pure (flip ($))~ recurse f v~ (pure (flip ($))~ recurse f u~pure Bin)
= [[ idiom interchange ]]

pure (flip ($))~ recurse f v~ (pure ($Bin)~ (pure (flip ($))~ recurse f u))
= [[ idiom composition ]]

pure (flip ($))~ recurse f v~ (pure (◦)~pure ($Bin)~pure (flip ($))~ recurse f u)
= [[ idiom homomorphism ]]

pure (flip ($))~ recurse f v~ (pure ((◦) ($Bin) (flip ($)))~ recurse f u)
= [[ sectioning ]]

pure (flip ($))~ recurse f v~ (pure (($Bin)◦flip ($))~ recurse f u)
= [[ idiom composition ]]

pure (◦)~ (pure (flip ($))~ recurse f v)~pure (($Bin)◦flip ($))~ recurse f u
= [[ idiom composition ]]

pure (◦)~pure (◦)~pure (flip ($))~ recurse f v~pure (($Bin)◦flip ($))~ recurse f u
= [[ idiom homomorphism ]]

pure ((◦) (◦) (flip ($)))~ recurse f v~pure (($Bin)◦flip ($))~ recurse f u
= [[ sectioning ]]

pure ((◦)◦flip ($))~ recurse f v~pure (($Bin)◦flip ($))~ recurse f u
= [[ idiom interchange ]]

pure ($(($Bin)◦flip ($)))~ (pure ((◦)◦flip ($))~ recurse f v)~ recurse f u
= [[ idiom composition ]]

pure (◦)~pure ($(($Bin)◦flip ($)))~pure ((◦)◦flip ($))~ recurse f v~ recurse f u
= [[ idiom homomorphism ]]

pure ((◦) ($(($Bin)◦flip ($))) ((◦)◦flip ($)))~ recurse f v~ recurse f u
= [[ sectioning ]]

pure (($(($Bin)◦flip ($)))◦ (◦)◦flip ($))~ recurse f v~ recurse f u
= [[ combinators: ($(($h)◦flip ($)))◦ (◦)◦flip ($) = flip h ]]

pure (flip Bin)~ recurse f v~ recurse f u

The final step with the hint ‘combinators’ just depends on the definitions of $ and flip: for any h,

(($(($h)◦flip ($)))◦ (◦)◦flip ($)) x y = ($(($h)◦flip ($))) ((◦) ($x)) y = (◦) ($x) (($h)◦flip ($)) y = (($x)◦ (($h)◦flip ($))) y
= ($x) ((($h)◦flip ($)) y) = (($h)◦flip ($)) y x = ($h) (flip ($) y) x = ($h) ($y) x = ($y) h x = h y x

Figure A.1. Proof of the claim in Section 6 that recurse collects the effects of traverse in the opposite order.

Here, we show the interesting case of the proof that B◦B is an idiom morphism. For brevity, define BB = B◦B, so that BB a = B (B a); we
will also use B and BB as subscripts on idiomatic operations.

BB mf ~BB BB ma
= [[ ~BB ]]

B ((pureB (flip ($))~B B ma)~B B mf )
= [[ pureB ]]

B ((B (pure (flip ($)))~B B ma)~B B mf )
= [[ ~B ]]

B (B (pure (flip ($))~ma~pure (flip ($)))~B B mf )
= [[ (1) below: pure (flip ($))~ma~pure (flip ($)) = pure (flip ($))~ma ]]

B (B (pure (flip ($))~ma)~B B mf )
= [[ ~B, BB ]]

BB (pure (flip ($))~mf ~ (pure (flip ($))~ma))
= [[ (2) below: (pure (flip ($))~mf )~ (pure (flip ($))~ma) = mf ~ma ]]

BB (mf ~ma)

Here’s subproof (1):

(pure (flip ($))~ma)~pure (flip ($))
= [[ idiom interchange ]]

pure (λ f → f (flip ($)))~ (pure (flip ($))~ma)
= [[ idiom composition ]]

pure (◦)~pure (λ f → f (flip ($)))~pure (flip ($)~ma)
= [[ idiom homomorphism ]]

pure ((◦) (λ f → f (flip ($))) (flip ($)))~ma
= [[ composition ]]

pure ((λ f → f (flip ($)))◦ (flip ($)))~ma
= [[ combinators: (λ f → f (flip ($)))◦ (flip ($)) = flip ($) ]]

pure (flip ($))~ma

The ‘combinators’ step is straightforward:

((λ f → f (flip ($)))◦ (flip ($))) h = (λ f → f (flip ($))) (flip ($) h) = (λ f → f (flip ($))) ($h) = ($h) (flip ($)) = flip ($) h

And here’s subproof (2):

(pure (flip ($))~mf )~ (pure (flip ($))~ma)
= [[ idiom composition ]]

pure (◦)~ (pure (flip ($))~mf )~pure (flip ($))~ma
= [[ idiom composition ]]

pure (◦)~pure (◦)~pure (flip ($))~mf ~pure (flip ($))~ma
= [[ idiom homomorphism ]]

pure ((◦) (◦) (flip ($)))~mf ~pure (flip ($))~ma
= [[ composition ]]

pure ((◦)◦flip ($))~mf ~pure (flip ($))~ma
= [[ idiom interchange ]]

pure (λ f → f (flip ($)))~ (pure ((◦)◦flip ($))~mf )~ma
= [[ idiom composition ]]

pure (◦)~pure (λ f → f (flip ($)))~pure ((◦)◦flip ($))~mf ~ma
= [[ idiom homomorphism ]]

pure ((◦) (λ f → f (flip ($))) ((◦)◦flip ($)))~mf ~ma
= [[ composition ]]

pure ((λ f → f (flip ($)))◦ (◦)◦flip ($))~mf ~ma
= [[ combinators: ((λ f → f (flip ($)))◦ (◦)◦flip ($)) = id ]]

pure id~mf ~ma
= [[ idiom identity ]]

mf ~ma

Figure A.2. Proof of the claim in Section 6 that B◦B : m .→ m is an idiom morphism.

